

Die Stellung der Jupitermonde – das BASIC-Programm JUMO

Die Jupitermonde zählen zu den am häufigsten beobachteten Himmelsobjekten, besonders in der Anfangsphase der Beobachter-tätigkeit. Im "Kalender für Sternfreunde" sind die Stellungen und Bewegungen der Monde graphisch dargestellt, wobei die Ablesegenauigkeit für eine Zuordnung ausreicht. In vielen Fällen ist aber eine genauere Kenntnis der Mondpositionen von Bedeutung. Dazu zählen zum Beispiel nahe Vorübergänge eines Mondes an einem anderen oder an Jupiter, Konjunktionen der Monde mit Sternen, Planeten oder anderen Himmelsobjekten. Aber auch im Rahmen eines Beobachtungsabendes mit Schülern oder für die eigene Tätigkeit bietet es sich an, besonders reizvolle Konstellationen im voraus zu kennen und gezielt zu beobachten. Dabei soll dieses Programm helfen. Es berechnet die Örter der vier großen Jupitermonde, wobei die Genauigkeit besser als 1 Winkelsekunde ist. Der Fehler ist also kleiner als ein Jupitermonddurchmesser. Somit sind auch näherungsweise Berechnungen von Jupitermondbedeckungen bzw. -verfinsterungen durch andere Monde möglich, wie Testrechnungen zu den gegenseitigen Bedeckungen und Verfinsterungen der vier großen Jupitermonde des Jahres 1985 (siehe "Kalender für Sternfreunde 1985" Seite 111f.) zeigten.

Das Programm JUMO (JupiterMonde) wurde auf einem Computer Z.1013 entwickelt. Da dieser Rechner nur auf sechs Stellen genau rechnet, habe ich das Bezugsäquinoktium von 1900.0 auf 1990.0 transformiert. Alle Absolutglieder c_0 in den Reihenentwicklungen der Form $= c_0 + c_1 D$ beziehen sich also auf den 0. 1. 1990 um 12h UT. So kann man auch mit sechs Stellen für einige Jahrzehnte um 1990 sehr gute Ergebnisse erhalten.

Für die Abschätzung der stellenbedingten Ungenauigkeit überlege man sich folgendes: Io bewegt sich pro Tag um 203 Grad weiter, das sind pro Jahr 74294 Grad, in 100 Jahren 7429399 Grad und in 1000 Jahren 74293991 Grad. Ein Grad Abweichung im Winkel hat einen Fehler von 0.1 Jupiterradius im Endergebnis zur Folge. Wenn man also die Stellung der Jupitermonde am 1. 1. des Jahres 1 auf 1 Winkelsekunde genau wissen möchte, so muß man mindestens einen zehnstelligen Rechner einsetzen. Andererseits heißt dies aber auch, daß für die Berechnung der Mondpositionen am 1. 1. des Jahres 2000 ein Computer mit sechs Stellen ausreicht, wenn die Genauigkeit wieder 1 Winkelsekunde betragen soll. Jeder Computerfreund sollte also die Grenzen seines Rechners kennen und ausschöpfen. Für die praktische Beobachtertätigkeit reichen 6 Stellen aus, wenn man die Differenz Standardepoche - Beobachtungszeit klein wählt; das gilt nicht nur für die Bewegung der Jupitermonde, sondern auch für den Erdmond, alle Planeten, Planetoiden usw.!

Ein weiteres Problem ist die fehlende Graphik des Z.1013. Hier mußte auf die Pseudographik zurückgegriffen werden. Sie liefert in 95 von 100 Fällen ein mit der echten Graphik vergleichbares Bild, nur in 5% aller Fälle liegen zwei Monde im gleichen Cursorfeld, so daß ein Mond nicht erkennbar ist. Diese Pseudographikroutine (Zeile 1500 - 1690) läuft nur auf dem Z.1013, mit kleinen Änderungen aber auch auf dem Z9001, KC 85/1 und KC 87. Auf allen anderen Rechnern ist diese Routine nicht lauffähig! Es muß also eine eigene, computer-spezifische Graphik geschrieben werden. Der Rechenteil (Zeile 600 bis 1170) ist dagegen auf fast allen Computern ungeändert lauffähig.

Auch sollte eine Umsetzung des Programms in die Programmiersprache PASCAL erwogen werden. JUMO wurde deshalb in Module unterteilt, welche nacheinander abgearbeitet werden und mit REM-Zeilen gekennzeichnet wurden. Auf GOSUB wurde vollkommen, auf GOTO weitestgehend verzichtet.

Das nachfolgend abgedruckte Programm stellt infolge der geringen Speicherkapazität des Z.1013 von knapp 5KByte nur ein lauffähiges Grundgerüst dar, welches erweitert werden sollte.

Als Ergänzungen sind denkbar:

- der "Blick von oben" auf die kreisförmigen Mondbahnen. Dies ist besonders für den Einsatz im Astronomieunterricht erwägenswert.
- die Berechnung der absoluten Koordinaten der Monde (Rektaszension und Deklination). Dabei ist die Neigung der Mondbahnen gegen die Bahn des Jupiters von $3^{\circ} 4'$ zu berücksichtigen (Variable DS; Neigung $N=0.05358 \cos(LA+.77667)$, LA, N in Radiant).
- die Ermittlung der gegenseitigen Jupitermondbedeckungen und -verfinsterungen. Die Radien der Jupitermonde betragen (in Jupiteräquatorradien): $R(1)=0.0253$, $R(2)=0.0218$, $R(3)=0.0367$, $R(4)=0.0336$.
- die Berechnung von Bedeckungen und Verfinsterungen durch Jupiter bzw. von Durchgängen vor Jupiter. Für die Beurteilung, ob Jupiter den Mond bedeckt oder umgekehrt, bildet man den Cosinus des Winkels Erde - Jupiter - Jupitermond (im Programm sind dies die Variablen U1 bis U4). Ist beispielsweise $\cos(U1)$ größer Null, so ist Mond 1 vor Jupiter, ist $\cos(U1)$ kleiner Null, so befindet sich Io hinter Jupiter, während er bei $\cos(U1)=0$ die größte Elongation erreicht hat. Mit den anderen Monden verfährt man analog.

Im Programm werden alle Winkel durch die in Zeile 140 definierte Funktion auf das Intervall 0...360 Grad reduziert und in Radiant umgewandelt. Damit wird berücksichtigt, daß die meisten Rechner das Winkelargument in Radiant verlangen.

Die Entfernungsangaben (Variablen X(i), Y(i), Ri; i=1, 2, 3, 4) beziehen sich alle auf den Jupiterradius am Äquator. Zur Umrechnung in das Winkelmaß benutze man folgende Beziehung: 1 Jupiterradius entspricht 98.44/DA Winkelsekunden. Wenn zum Beispiel Kallisto 25 Jupiterradien entfernt und Jupiteropposition ist (DA=4.2), so steht Kallisto 586 Winkelsekunden von Jupiter entfernt.

Beim nachfolgenden Listing ergänzen bzw. ändern Sie bitte, wenn Sie Besitzer des Z.1013 in der 16K-Grundvariante sind, folgende Zeilen (durch sie wird der Zeichenkettenspeicher gekürzt):

```

130 CLEAR 7
2000 CLEAR 255:END
1260 PRINTAT(0,19);" um",STR$(SD),"h",STR$(MI),"m  "

```

LISTING FUER Z.1013

```
10 REM Berechnung und graphische Ausgabe der Stellung der
20 REM Jupitermonde
100 REM*****
110 REM*** INITIALISIERUNG ***
120 REM*****
140 DEF FN Y(Z)=PI*Z/180-INT(Z/360)*(PI+PI)
150 DIM X(4),Y(4),P(4)
160 Q=PI/180
170 ZZ=32:REM ZEILENLANGE, GILT NUR FUER Z.1013
200 REM*****
210 REM*** DATUMSEINGABE ***
220 REM*****
230 CLS
240 PRINT" Berechnung der gegenseitigen"
250 PRINT" Stellung der Jupitermonde"
260 PRINT"-----"
270 PRINT
280 PRINT" Programmierer: "
290 PRINT"           Jens Brunzendorf"
300 PRINT"           Gruenauer Allee 6"
310 PRINT"           Leipzig, 7062"
320 PRINT:PRINT
330 PRINT" Das Programm basiert auf den"
340 PRINT" Formeln von Jean Meeus' Buch"
350 PRINT" ";CHR$(34);"Astronomical Formulae for"
360 PRINT" Calculators, second Edition";CHR$(34)
370 PRINT:PRINT
380 PRINT" Geben Sie nun das Datum ein !"
390 INPUT"           Tag :";TG
400 INPUT"           Monat :";MN
410 INPUT"           Jahr :";JR
420 PRINT
430 INPUT" Stunde in MEZ :";SD
440 INPUT"           Minuten :";MI
450 CLS
500 REM*****
510 REM*** TAGE SEIT 1990.0 ***
520 REM*****
530 M=MN+1;J=JR:IF MN<3 THEN M=MN+13;J=JR-1
540 D=INT(365.25*(J-1988))+INT(30.6002*M)+TG
550 D=D-INT(J/100)+INT(J/400)-61.5
560 D=D+SD/24+MI/1440-1/24
570 D=D-715:REM D=0 fuer 0. 1. 1990, 12 h UT
580 DT=.002853*JR*JR-10.49778*JR+9651:D=D+DT/24/3600:REM UT -> ET
600 REM*****
610 REM*** BEGINN DER RECHNUNG ***
620 REM*****
630 V=171.31+.00111587*D:V=FN Y(V)
640 M=355.150+.9856003*D:M=FN Y(M)
650 N=76.342+.0030853*D+.33*SIN(V):N=FN Y(N)
660 J=7.4104+.9025179*D-.33*SIN(V):J=FN Y(J)
670 A=1.916*SIN(M)+.02*SIN(M+M):A=FN Y(A)
680 B=5.552*SIN(N)+.167*SIN(N+N):B=FN Y(B)
690 K=J+A-B
700 R=1.00014-.01672*COS(M)-.00014*COS(M+M)
710 RR=5.20067-.25192*COS(N)-.0061*COS(N+N)
720 DA=SQR(R*R+RR*RR-2*R*RR*COS(K))
730 W=R/DA*SIN(K)
740 W=ATN(W/SQR(1-W*W))
```

```
750 LA=89.25+.083091*D+.33*SIN(Y):LA=FN Y(LA)+B
760 DS=.05358*SIN(LA+.77667)
770 DE=DS-.0375246*SIN(W)*COS(LA+.419)
780 DE=DE-.022864*(RR-DA)*SIN(LA-1.73486)/DA
790 X=D-DA/173
800 U1=115+203.4*X:U1=FN Y(U1)+0*(.2673+.005863*X)+W-B
810 U2=217+101.2*X:U2=FN Y(U2)+0*(.4552+.0916323*X)+W-B
820 U3=358+50.2*X:U3=FN Y(U3)+0*(.5475+.0345169*X)+W-B
830 U4=166+21.4*X:U4=FN Y(U4)+0*(.26776+.0879802*X)+W-B
840 G=59.15+50.3*X:G=FN Y(G)+0*X*.010674
850 H=91.66+21.5*X:H=FN Y(H)+0*X*.069229
860 R1=5.9061-.0244*COS(2*(U1-U2))
870 R2=9.3972-.0889*COS(2*(U2-U3))
880 R3=14.9894-.0227*COS(G)
890 R4=26.3649-.1944*COS(H)
900 U1=.472*SIN(2*(U1-U2))*Q+U1
910 U2=1.073*SIN(2*(U2-U3))*Q+U2
920 U3=.174*SIN(G)*Q+U3
930 U4=.845*SIN(H)*Q+U4
940 X(1)=R1*SIN(U1)
950 X(2)=R2*SIN(U2)
960 X(3)=R3*SIN(U3)
970 X(4)=R4*SIN(U4)
980 Y(1)=R1*COS(U1)*SIN(DE)
990 Y(2)=R2*COS(U2)*SIN(DE)
1000 Y(3)=R3*COS(U3)*SIN(DE)
1010 Y(4)=R4*COS(U4)*SIN(DE)
1100 REM*****
1110 REM*** ENDE DER BERECHNUNG ***
1120 REM*****
1130 REM X...westlicher Abstand in Jupiterradien,
1140 REM nach Osten negativ.
1150 REM Y...suedlicher Abstand in Jupiterradien,
1160 REM nach Norden negativ.
1170 REM Jupiterradius am Aequator gemessen.
1200 REM*****
1210 REM*** AUSGABE ***
1220 REM*****
1230 WINDOW
1240 PRINT:PRINT
1250 PRINTAT(0,0);"Datum:",STR$(TG),".",STR$(MN),".",STR$(JR)," "
1260 PRINTAT(0,19);" um",STR$(SD),"h",STR$(MI),"m MEZ "
1270 PRINT"Abstaende in Jupiterradien:"
1280 PRINT
1290 PRINT"          waagerecht   senkrecht"
1300 PRINT"          Io"
1310 PRINT"          Europa"
1320 PRINT"          Ganymed"
1330 PRINT"          Kallisto"
1340 FOR I=1 TO 4:Q=0:IF ABS(X(I))<10 THEN Q=1
1350 IF ABS(X(I))<1 THEN Q=2
1360 PRINTAT(I+5,10+Q);" ",INT(X(I)*1000+.5)/1000," "
1370 PRINTAT(I+5,23);" ",INT(Y(I)*1000+.5)/1000," "
1380 NEXT
1390 PRINT
1400 PRINT"(West und Sued positiv, Ost und"
1410 PRINT" Nord negativ gerechnet.)"
1500 REM*****
1510 REM*** BEGINN DER Z.1013-GRAPHIK ***
1520 REM*****
```

```
1530 FOR I=1 TO 4
1540 YY=-Y(I)/2+.5
1550 XX=INT(YY)*ZZ
1560 YY=YY-INT(YY)
1570 YY=INT(4*YY)
1580 X=-X(I)/2+15.5
1590 P(I)=INT(4*X-4*INT(X))+4*YY+208
1600 X(I)=INT(X)+XX-4480
1610 NEXT
1620 PRINT:PRINT"Anblick im umkehrenden Fernrohr "
1630 PRINTTAB(14)"Sued"
1640 PRINT:PRINT"West                Ost";
1650 PRINTSPC(6*ZZ)
1660 FOR I=1 TO 4:POKE X(I),P(I):POKE X(I)+3*ZZ,48+I
1670 NEXT
1680 POKE -4465,207
1690 PRINTAT(25,14);"Nord":PRINT:PRINT
1700 REM*****
1710 REM*** Veraenderung des Datums ***
1720 REM*****
1730 PRINT"1-0... +1h bis +10h"
1740 PRINT"<- ... -1min,          ->... +1min"
1750 PRINT"ENT... +10min,  SPACE... -30min";
1760 PRINT"  N... neues Datum, E... Ende".
1770 POKE -4128,32
1780 A$=INKEY$:IF A$="" THEN1780
1790 IF A$="N" THEN RUN
1800 X=0:IF VAL(A$)>0 THEN X=60*VAL(A$)
1810 XX=ASC(A$)
1820 IF XX=13 THEN X=10
1830 IF XX=8  THEN X=-1
1840 IF XX=9  THEN X=1
1850 IF XX=32 THEN X=-30
1860 IF XX=48 THEN X=600
1870 IF A$="E" THEN CLS:GOTO2000
1880 MI=MI+X:D=D+X/1440
1890 XX=INT(MI/60)
1900 SD=SD+XX:MI=MI-60*XX
1910 XX=INT(SD/24)
1920 TG=TG+XX:SD=SD-24*XX
1930 Q=PI/180
1940 GOTO610
2000 END
OK
```

Für die Anpassung dieses Listings an den Z9001, KC 85/1 und KC 87 ändere man bitte folgende Zeilen:

```
170 ZZ=40
1360 PRINTAT(I+4,10+Q);" ",INT(X(I)±1000+.5)/1000," "
1370 PRINTAT(I+4,23);" ",INT(Y(I)±1000+.5)/1000," "
1620 entfällt
1690 PRINT
1760 entfällt
1770 POKE -4168,32
```

FÜR ALLE ANDEREN COMPUTERTYPEN KÖNNEN DIE ZEILEN 1500 BIS 1680 SOWIE ZEILE 1770 NICHT VERWENDET WERDEN !!

T E S T B E I S P I E L

Variablenwerte für den 31. 1. 1978 um 21h 30m MEZ

Ein Rechner mit acht Stellen liefert folgende Werte:

D =-4349.6453	DT=48.708008
V =2.9052116	M = .49124596
N =1.3094804	J = .72780937
A = .016065101	B = .095065892
K = .64880858	W = .13559704
LA=1.6294007	
DS= .035950803	DE= .038689404
U1=4.3405533	U2=-1.4836857
U3= .30422331	U4=-1.007401
G =1.8201001	H =-2.2232697
R1=5.8907233	R2= 9.4774634
R3=14.995001	R4=26.482931

X(1)= -5.48816276	Y(1)=-.0827845282
X(2)= -9.4415274	Y(2)= .0318931474
X(3)= 4.49178602	Y(3)= .553369131
X(4)=-22.389906	Y(4)= .547065894 .

Ein Rechner mit sechs Stellen liefert

X(1)= -5.48863	Y(1)=-.0827366
X(2)= -9.44167	Y(2)= .0318393
X(3)= 4.4848	Y(3)= .553454
X(4)=-22.3888	Y(4)= .547135 .

Kann man dagegen einen sechzehnstelligen Rechner einsetzen, so erhält man:

X(1)= -5.489136777144734	Y(1)=-.08268683181161223
X(2)= -9.441368182232582	Y(2)= .03197310815280944
X(3)= 4.490956032720823	Y(3)= .5533787601584601
X(4)=-22.38951977604387	Y(4)= .5470889194785637 .

Im Rahmen der Formelungenauigkeit liefern alle drei Rechner das gleiche Ergebnis, der größte Unterschied beträgt bei X(3) 0.006 Jupiterradien, dem entsprechen 0.3 Winkelsekunden.

In den vergangenen Veröffentlichungen wurden einige Fehler entdeckt, die wir zu korrigieren bitten:

Veröffentlichung 2 : Berechnung von Ephemeriden...

Seite 3

Die Formeln für die Gaußschen Konstanten lauten korrekt:

$$Q_y = -\sin(\epsilon) \pm H^2 - \cos(\epsilon) \pm H^4 \quad (36)$$

$$Q_z = \cos(\epsilon) \pm H^2 - \sin(\epsilon) \pm H^4 \quad (37)$$

Veröffentlichung 5 : Reduktionsrechnung

Seite 1

In der Tabelle wurden Bezeichnungen vertauscht. Es muß heißen:

wahrer Ort	Nutation	wahrer Ort
mittlerer Ort zum Beobachtungszeitpunkt	Präzession	mittlerer Ort zum Beobachtungszeitpunkt
mittlerer Ort zur Epoche		mittlerer Ort zur Epoche

Seite 6: Die Zeilen 5500 bis 5595 streichen.

Seite 7: 6040 LET EE=INT((BB-DD)/30.0061)
6050 LET TA=BB-DD-INT(30.0061*EE)+F1

Seite 9: 6791 IF K\$<>Q\$ THEN GOTO 6515
ergänzen.

Seite 10: 7890 IF (WS<0) OR (WS>=60) THEN GOTO 7880

Seite 11: 8784 IF (MI<0) OR (MI>=60) THEN GOTO 8770

Seite 12: Zeilen 8835 und 8840 vertauschen.

Veröffentlichung 6 : Sonne nach NEWCOMB

Seite 9: In der Datei "SUNNEW.DAT" ist der letzte Venustern (1.Block, letzte Zeile) fehlerhaft. Die Zeile muß lauten:

0.154 359.6 5 -8

Anschriften der Autoren:

Jens Brunzendorf
Grünauer Allee 6
Leipzig, 7062

Uwe Pilz
Lazarusstr. 23
Leipzig, 7024

Die Autoren sind Mitglied im

Arbeitskreis "Numerische Astronomie"
(Leiter: Dipl.-Phys. L. Ehrenberg)
PSF 29
Schkeuditz
7144